; This script need to use a special Inno Setup version located here : http://jrsoftware.org/isdl.php#qsp


#pragma option -v+
#pragma verboselevel 9
#define Debug


#if RELEASE=="Professional"
  #define SOFTWARE_NAME "AnyDbTest Professional Edition" 
  #define OUTPUT_FILE_NAME "AnyDbTest_Prof"
#else 
  #define SOFTWARE_NAME "AnyDbTest Community Edition"
  #define OUTPUT_FILE_NAME "AnyDbTest_Community"  
#endif

#define SOFTWARE_PUBLISHER_SHORT_NAME "AnyDbTest"
#define SOFTWARE_COPYRIGHTS "2011 AnyDbTest Technology"
#define SOFTWARE_PUBLISHER "AnyDbTest Technology"
#define SOFTWARE_PRODUCT_PAGE "http://AnyDbTest.codeplex.com/" 
#define SOFTWARE_DOWNLOAD_PAGE "http://anydbtest.codeplex.com/releases/"
#define SOFTWARE_SUPPORT_PAGE "mailto:AnyDbTestSupport@gmail.com"
#define SOFTWARE_SUPPORT_MAIL "AnyDbTestSupport@gmail.com"
#define SOFTWARE_VERSION GetFileVersion("..\dist\AnyDbTest.exe") 
 

[Setup]
SourceDir=..\dist
OutputDir=..\installer

; NOTE: The value of AppId uniquely identifies this application.
; Do not use the same AppId value in installers for other applications.
; (To generate a new GUID, click Tools | Generate GUID inside the IDE.)
AppId={{D7DEDD5E-321A-4215-B843-90E946735F64}

AppVerName={#SOFTWARE_NAME} {#SOFTWARE_VERSION}
AppVersion={#SOFTWARE_VERSION}
VersionInfoVersion={#SOFTWARE_VERSION}
VersionInfoTextVersion={#SOFTWARE_VERSION}
VersionInfoCompany={#SOFTWARE_PUBLISHER}
VersionInfoDescription=Handy database unit testing and performance testing tool
VersionInfoCopyright={#SOFTWARE_COPYRIGHTS}

AppName={#SOFTWARE_NAME} 
AppCopyright={#SOFTWARE_COPYRIGHTS}
AppPublisher={#SOFTWARE_PUBLISHER}
AppPublisherURL={#SOFTWARE_PRODUCT_PAGE}
AppSupportURL={#SOFTWARE_SUPPORT_PAGE}
AppUpdatesURL={#SOFTWARE_DOWNLOAD_PAGE}
AppComments=Thanks using {#SOFTWARE_NAME}
AppContact={#SOFTWARE_SUPPORT_MAIL}

DefaultDirName={pf}\{#SOFTWARE_PUBLISHER_SHORT_NAME}\{#SOFTWARE_NAME}\{#SOFTWARE_VERSION}
DefaultGroupName={#SOFTWARE_NAME}\{#SOFTWARE_VERSION}
OutputBaseFilename={#OUTPUT_FILE_NAME}
LicenseFile=EULA\AnyDbTest_EULA.rtf

AllowNoIcons=yes
Compression=lzma2/ultra
InternalCompressLevel=ultra
SolidCompression=yes 



[Languages]
Name: "english"; MessagesFile: "compiler:Default.isl"

[Tasks]
Name: "desktopicon"; Description: "{cm:CreateDesktopIcon}"; GroupDescription: "{cm:AdditionalIcons}"; Flags: unchecked

[Files]
Source: "*" ;  DestDir: "{app}"; Flags: ignoreversion recursesubdirs createallsubdirs

[Icons]
Name: "{group}\AnyDbTest GuiRunner"; Filename: "{app}\AnyDbTest.exe"
Name: "{group}\DB Connection Builder"; Filename: "{app}\DbConnectionBuilder.exe"
Name: "{group}\AnyDbTest ConsoleRunner"; Filename: "{app}\ConsoleRunner.exe"
;;Name: "{group}\Convert XML Data to Excel"; Filename: "{app}\Xml2Excel.exe"
Name: "{group}\Visit Web Site"; Filename: "{app}\help\website.url"

Name: "{userdesktop}\AnyDbTest GuiRunner"; Filename: "{app}\AnyDbTest.exe"; Tasks: desktopicon

[Run]
Filename: "{app}\AnyDbTest.exe"; Description: "{cm:LaunchProgram,AnyDbTest GuiRunner}"; Flags: nowait postinstall skipifsilent

